# POGLAVJE II

## II.5 Reševanje diferencialne enačbe prečnega pomika lastnega nihanja

*1. Numerično iskanje lastnih frekvenc za konzolni nosilec*
*2. Analitično iskanje lastnih nihajnih oblik za konzolni nosilec*

### 1. Numerično iskanje lastnih frekvenc za konzolni nosilec

Splošna rešitev diferencialne enačbe je bila poiskana že v prejšnjem podpoglavju. Namesto, da se sedaj za analizo drugačne konstrukcije ponovi ves postopek njene pridobitve oz. izpeljave, se lahko iz prejšnje vaje rešitev kar privzame (prekopira).
Odpre se datoteka (notebook), kjer je shranjena prejšnja vaja (poglavje II.4), in se poišče izhodna vrstica, kjer je zapisan rezultat, kurzor pa se postavi na konec vrstice:

```
Simplify[%] /. {C[2] + C[4] → C[2], -C[2] + C[4] → C[4]}

C[1] Cos[b x] + C[2] Cosh[b x] +
 C[3] Sin[b x] + C[4] Sinh[b x]
```

Izbrana izhodna vrstica se shrani v odložišče (Clipboard) s *Ctrl+C*, in stara datoteka se lahko zapre (ker ni več potrebna). Nato se odpre nova datoteka, v kateri se bo izvajala nadaljnja analiza, in vanjo se shranjeni niz odloži s *Ctrl+V*. Niz se pri tem shrani v svoji originalni formi oz. definiciji, torej kot izhodni niz (in ga zato ni mogoče izvesti) ter ga je zato najprej potrebno pretvoriti v vhodnega. Zato se klikne oz. označi se konec vrstice:

```
C[1] Cos[b x] + C[2] Cosh[b x] + C[3] Sin[b x] + C[4] Sinh[b x]
```

z kombinacijo tipk *Ctrl+Shift+I* ali ukazi *Cell/Convert To/InputForm* pa se vrstica iz izhodne spremeni v vhodno vrstico:

```
C[1]*Cos[b*x] + C[2]*Cosh[b*x] + C[3]*Sin[b*x] +
C[4]*Sinh[b*x]
```

ki navidezno zgolj spremeni obliko. Za dokončanje dela se kurzor postavi na začetek vrstice:

```
C[1]*Cos[b*x] + C[2]*Cosh[b*x] + C[3]*Sin[b*x] +
C[4]*Sinh[b*x]
```

in natipka se ime funkcije in argument. *Mathematica* avtomatično tvori novo, končno vhodno vrstico:

```
w[x_]=C[1]*Cos[b*x] + C[2]*Cosh[b*x] + C[3]*Sin[b*x] +
C[4]*Sinh[b*x]
```

ki jo je sedaj mogoče tudi izvesti:

```
In[1]:= w[x_]=C[1]*Cos[b*x] + C[2]*Cosh[b*x] + C[3]*Sin[b*x] +
        C[4]*Sinh[b*x]

Out[1]= C[1] Cos[b x] + C[2] Cosh[b x] + C[3] Sin[b x] + C[4] Sinh[b x]
```

Iz prejšnje datoteke skopirano vrstico pa je sedaj mogoče brez škode tudi izbrisati.

Splošna funkcija ima enake konstante kot v prejšnji vaji (in lahko zato tudi to vrstico po potrebi od tam kar skopiramo):

```
In[2]:= neznanke = Table[C[i], {i, 4}]

Out[2]= {C[1], C[2], C[3], C[4]}
```

Za levo vpeti konzolni nosilec dolžine L pa veljajo naslednji robni pogoji:

- **Pogoji za levo vpeto konzolo**

```
In[3]:= pogoji = {w[0] == 0, w'[0] == 0, w''[L] == 0, w'''[L] == 0};
```

ki spet predstavljajo štiri linearne homogene enačbe s petimi neznankami:

```
In[4]:= pogoji
```

Out[4]=

$$\{C[1]+C[2]==0,\ b\,C[3]+b\,C[4]==0,$$
$$-b^2\,C[1]\,\mathrm{Cos}[b\,L]+b^2\,C[2]\,\mathrm{Cosh}[b\,L]-b^2\,C[3]\,\mathrm{Sin}[b\,L]+b^2\,C[4]\,\mathrm{Sinh}[b\,L]==0,$$
$$-b^3\,C[3]\,\mathrm{Cos}[b\,L]+b^3\,C[4]\,\mathrm{Cosh}[b\,L]+b^3\,C[1]\,\mathrm{Sin}[b\,L]+b^3\,C[2]\,\mathrm{Sinh}[b\,L]==0\}$$

Ker je že znano, da klasično reševanje sistema linearnih homogenih enačb pripelje zgolj do inženirsko neuporabne trivialne rešitve, se enačbe zapišejo v matrični obliki, da se bo lahko izračunala determinanta:

```
In[5]:= matrika =
          Table[Coefficient[pogoji[[i, 1]], neznanke[[j]]],
           {i, 1, 4}, {j, 1, 4}];
```

Rezultat ima naslednjo obliko:

```
In[6]:= MatrixForm[matrika]
```

Out[6]//MatrixForm=

$$\begin{pmatrix} 1 & 1 & 0 & 0 \\ 0 & 0 & b & b \\ -b^2\,\mathrm{Cos}[b\,L] & b^2\,\mathrm{Cosh}[b\,L] & -b^2\,\mathrm{Sin}[b\,L] & b^2\,\mathrm{Sinh}[b\,L] \\ b^3\,\mathrm{Sin}[b\,L] & b^3\,\mathrm{Sinh}[b\,L] & -b^3\,\mathrm{Cos}[b\,L] & b^3\,\mathrm{Cosh}[b\,L] \end{pmatrix}$$

Izračuna se še determinanta:

■ **Lastna frekvenca**

```
In[7]:= determinanta = Det[matrika]

Out[7]= -b^6 Cos[b L]^2 - 2 b^6 Cos[b L] Cosh[b L] -
         b^6 Cosh[b L]^2 - b^6 Sin[b L]^2 + b^6 Sinh[b L]^2
```

ki pa je sedaj nekoliko kompleksnejša in zato se takoj poskusi poenostaviti:

```
In[8]:= Simplify[%]

Out[8]= -2 b^6 (1 + Cos[b L] Cosh[b L])
```

Rešitve te enačbe so vrednosti koeficienta b (pete neznanke), ki je povezan z lastno frekvenco konstrukcije. Ena izmed očitnih rešitev je seveda b≡0, ki pa, kot je bilo že pokazano v prejšnjem podpoglavju, ne predstavlja pomika. Ker je potem jasno, da uporabna rešitev sledi, ko je izraz v oklepaju enak nič, se zato iz izraza eleminirajo nepotrebni členi:

```
In[9]:= determinanta = Simplify[determinanta / -2 / b^6]

Out[9]= 1 + Cos[b L] Cosh[b L]
```

Enačba se najprej poskusi rešiti direktno:

```
In[10]:= Solve[% == 0, b]

         Solve::tdep :
          The equations appear to involve the variables to be
            solved for in an essentially non-algebraic way. More…

Out[10]= Solve[1 + Cos[b L] Cosh[b L] == 0, b]
```

kar se izkaže kot neuspešno in zato se enačba najprej preuredi z pretvorbo izraza iz trigonometričnih funkcij v eksponentne z ukazom *TrigToExp[izraz]*:

```
In[11]:= TrigToExp[determinanta]
```

Out[11]= $1 + \frac{1}{4} e^{(-1-i) b L} + \frac{1}{4} e^{(-1+i) b L} + \frac{1}{4} e^{(1-i) b L} + \frac{1}{4} e^{(1+i) b L}$

in poskuša rešiti ponovno:

```
In[12]:= Solve[% == 0, b]

Solve::tdep :
 The equations appear to involve the variables to be
   solved for in an essentially non-algebraic way. More…
```

Out[12]= Solve[

$$1+\frac{1}{4}e^{(-1-i)bL}+\frac{1}{4}e^{(-1+i)bL}+\frac{1}{4}e^{(1-i)bL}+\frac{1}{4}e^{(1+i)bL}==0,\ b]$$

a tudi tokrat brezuspešno.

Zaradi tega se enačna poskusi rešiti numerično. Ker kot njen argument nastopa produkt dveh spremenljivk, se namesto njiju vpelje nova, da se dobi enačba zgolj z eno neznanko. Tako se npr. vpelje:
$\xi = b \cdot L$ ,
kar vodi do preglednejše oblike:

```
In[13]:= funkcija = determinanta /. b L → ξ

Out[13]= 1 + Cos[ξ] Cosh[ξ]
```

Kadar se iščejo ničle neke enačbe, je lahko v veliko pomoč izris funkcije, saj omogoča vsaj približno lokacijo ničel. Ker je lahko lastna frekvenca zgolj pozitivna, je spodnja meja spremenljivke $\xi$ enaka nič, zgornja meja pa se lahko poljubno izbere (po potrebi s poskušanjem):

```
In[14]:= Plot[funkcija, {ξ, 0, 10}];
```




Izkaže se, da je bila izbira uspešna, saj je iz slike mogoče locirati vsaj tri ničle (amplituda v področju med 0 in 2 je premajhna, da bi se lahko natančno določilo dogajanje v tem intervalu). Vsaka izmed njih se analizira posebej.

Poseben izris področja med 0 in 3 potrdi, da je v tem intervalu res zgolj ena ničla:

### ■ 1. frekvenca

```
In[15]:= Plot[funkcija, {ξ, 0, 3}];
```




Njena natačna lokacija se lahko poišče z ukazom *FindRoot[funkcija,{spremenljivka,približna lokacija iskane ničle}]*:

```
In[16]:= FindRoot[funkcija, {ξ, 2}]

Out[16]= {ξ → 1.8751}
```

izračunana vrednost pa se shrani npr. v spremenljivko *ksi*:

```
In[17]:= ksi = ξ /. %[[1]]

Out[17]= 1.8751
```

Iz zveze med konstanto b in lastno krožno frekvenco ω se le ta izrazi kot funkcija b (izraz je splošen, saj se bo uporabil še pri izračunu višjih frekvenc):

In[18]:= omega = Solve$\left[b^4 == \frac{\omega^2 m}{EI}, \omega\right]$

Out[18]= $\left\{\left\{\omega \to -\frac{b^2\sqrt{EI}}{\sqrt{m}}\right\}, \left\{\omega \to \frac{b^2\sqrt{EI}}{\sqrt{m}}\right\}\right\}$

Za izračun prve lastne krožne frekvence se uporabi pozitivna rešitev, upošteva pa se tudi izračunana vrednost koeficienta ξ:

```
In[19]:= ω1 = ω /. omega[[2]] /. b → ksi / L
```

Out[19]= $\frac{3.51602\sqrt{EI}}{L^2\sqrt{m}}$

Uspešnost oz. kvaliteta rezultata se lahko preveri z izračunom determinante, ki mora biti enaka nič:

```
In[20]:= determinanta /. {ω → ω1, b → ksi /L}

Out[20]= 2.22045 × 10^-16
```

kar potrdi zadostno natančnost rešitve, saj gre za minimalno odstopanje, ki je posledica numerične analize.

Postopek je sedaj mogoče smiselno uporabiti še za izračun druge in (po potrebi) vseh ostalih višjih lastnih krožnih frekvenc:

- **2. frekvenca**

```
In[21]:= Plot[funkcija, {ξ, 4, 6}];
```




```
In[22]:= FindRoot[funkcija, {ξ, 4.5}]

Out[22]= {ξ → 4.69409}

In[23]:= ksi = ξ /. %[[1]]

Out[23]= 4.69409

In[24]:= ω2 = ω /. omega[[2]] /. b → ksi /L
```

$$\text{Out[24]=}\ \frac{22.0345\sqrt{EI}}{L^2\sqrt{m}}$$

```
In[25]:= determinanta /. {ω → ω2, b → ksi /L}

Out[25]= 2.36478 × 10^-14
```

Tako je mogoče izračunati še tretjo

- **3. frekvenca**

In[26]:= `Plot[funkcija, {ξ, 7, 8.5}];`




In[27]:= `FindRoot[funkcija, {ξ, 7.8}]`

Out[27]= $\{\xi \to 7.85476\}$

In[28]:= `ksi = ξ /. %[[1]]`

Out[28]= 7.85476

In[29]:= `ω3 = ω /. omega[[2]] /. b → ksi / L`

Out[29]= $\frac{61.6972\sqrt{EI}}{L^2\sqrt{m}}$

In[30]:= `determinanta /. {ω → ω3, b → ksi / L}`

Out[30]= $-8.70415\times10^{-14}$

ter še vse preostale lastne frekvence (faze izračuna niso prikazane), ki se npr. shranijo v vektor *omege* (zaradi podobnosti s predefiniranim imenom *omega Mathematica* javi opozorilo o morebitni napaki):

In[46]:= `omege = {ω1, ω2, ω3, ω4, ω5, ω6}`

```
General::spell1 :
 Possible spelling error: new symbol name "omege" is
   similar to existing symbol "omega". More…
```

Out[46]= $\left\{\frac{3.51602\sqrt{EI}}{L^2\sqrt{m}}, \frac{22.0345\sqrt{EI}}{L^2\sqrt{m}}, \frac{61.6972\sqrt{EI}}{L^2\sqrt{m}}, \frac{120.902\sqrt{EI}}{L^2\sqrt{m}}, \frac{199.86\sqrt{EI}}{L^2\sqrt{m}}, \frac{298.556\sqrt{EI}}{L^2\sqrt{m}}\right\}$

## 2. Analitično iskanje lastnih nihajnih oblik za konzolni nosilec

Pri iskanju lastnih nihajnih oblik se je spet potrebno vrniti k linearnemu sistemu homogenih enačb. Ker je že bilo zagotovljeno, da je vrednost *b* takšna, da je determinanta enaka nič, je potrebno še določiti koeficiente C[1], C[2], C[3] in C[4]. Ker pa v bistvu nastopa neskočna množica vrednosti *b*, nastopa tudi neskončna množica rešitev za konstante C[1], C[2], C[3] in C[4] (za vsak *n* ena). Zaradi tega je smiselno enačbe, zapisane v vektorju *pogoji*, čim bolje analizirati na čim splošnejšem nivoju.

Tako je mogoče opaziti, da je mogoče drugo, tretjo in četrto enačbo oz. vrstico pokrajšati z b, $b^2$ oz. $b^3$ (slično kot je to bilo narejeno v prejšnjem poglavju, II.4), vendar sedaj prvi dve enačbi vseeno zajameta vse štiri konstante (in ne zgolj prvih dveh, kar je v prejšnjem podpoglavju omogočilo razdelitev iskanja rešitve na dva sistema).
Zato je potrebno vsako enačbo reševati posebej in je potrebno vse konstante spet izraziti z eno, saj gre za enoparametrično množico rešitev.

Iz prve enačbe (ki vsebuje zgolj dve neznanki) se npr. izrazi konstanta C[2] kot funkcija C[1]:

### ■ Lastne nihajne oblike

```
In[47]:= prva = Solve[pogoji[[1]], C[2]]

Out[47]= {{C[2] → -C[1]}}
```

rezultat pa se upošteva v zapisanih pogojih (v tretji in četrti enačbi se namesto C[2] upošteva ustrezna zveza z C[1], prva enačba pa se dejansko izvede):

```
In[48]:= pogoji = pogoji /. %[[1]]

Out[48]= {True, b C[3] + b C[4] == 0,
  -b^2 C[1] Cos[b L] - b^2 C[1] Cosh[b L] -
    b^2 C[3] Sin[b L] + b^2 C[4] Sinh[b L] == 0,
  -b^3 C[3] Cos[b L] + b^3 C[4] Cosh[b L] +
    b^3 C[1] Sin[b L] - b^3 C[1] Sinh[b L] == 0}
```

Nato se npr. iz druge enačbe (ki tudi vsebuje zgolj dve neznanki) izrazi C[4] s C[3]:

```
In[49]:= druga = Solve[pogoji[[2]], C[4]]

Out[49]= {{C[4] → -C[3]}}
```

ki se prav tako upošteva v sistemu pogojev (v tretji in četrti vrstici):

```
In[50]:= pogoji = pogoji /. %[[1]]

Out[50]= {True, True, -b^2 C[1] Cos[b L] - b^2 C[1] Cosh[b L] -
    b^2 C[3] Sin[b L] - b^2 C[3] Sinh[b L] == 0,
  -b^3 C[3] Cos[b L] - b^3 C[3] Cosh[b L] +
    b^3 C[1] Sin[b L] - b^3 C[1] Sinh[b L] == 0}
```

Sedaj je mogoče iz tretje ali četrte enačbe še eno izmed preostalih dveh konstant izraziti z drugo (postopek določitve konstant je identičen postopku pri Gaussovi eliminaciji, ko je sistem že preveden v trikotno obliko), npr.:

```
In[51]:= tretja = Solve[pogoji[[3]], C[3]]

Out[51]= {{C[3] → (-C[1] Cos[b L] - C[1] Cosh[b L]) / (Sin[b L] + Sinh[b L])}}
```

Vse, s konstanto C[1] izražene konstante, je sedaj potrebno upoštevati v funkciji lastne nihajne oblike. Ker rešitev za C[4] vsebuje še C[3], se najprej upoštevata zvezi za C[2] (shranjena pod imenom *prva*) in C[4] (shranjena pod imenom *druga*), na koncu pa še dodatno rešitev za C[3] (shranjena pod imenom *tretja*):

```
In[52]:= w[x_] = w[x] /. {prva[[1, 1]], druga[[1, 1]]} /.
           tretja[[1, 1]]

Out[52]= C[1] Cos[b x] - C[1] Cosh[b x] +
           ((-C[1] Cos[b L] - C[1] Cosh[b L]) Sin[b x]) / (Sin[b L] + Sinh[b L]) -
           ((-C[1] Cos[b L] - C[1] Cosh[b L]) Sinh[b x]) / (Sin[b L] + Sinh[b L])
```

Izraz se lahko tudi poskusi poenostaviti z ukazom *Simplify[izraz]*, vendar brez posebnega uspeha:

```
In[53]:= Simplify[w[x]]

Out[53]= (C[1] (Cos[b x] Sin[b L] -
              Cos[b L] Sin[b x] + Cos[b x] Sinh[b L] -
              Cosh[b x] (Sin[b L] + Sinh[b L]) + Cos[b L] Sinh[b x] +
              Cosh[b L] (-Sin[b x] + Sinh[b x]))) /
           (Sin[b L] + Sinh[b L])
```

Pred izrisom posameznih lastni nihajnih oblik je morda smiselno funkcije še urediti tako, da se upošteva, da se v bistvu posamezne nihajne oblike razlikujejo zgolj po koeficientu *b*, ki pa je funkcija pripadajajoče lastne frekvence, in zato se najprej zapiše ta zveza:

In[54]:= $\mathbf{B} = \sqrt{\sqrt{\dfrac{\omega^2\,\mathrm{m}}{\mathrm{EI}}}}$

Out[54]= $\left(\dfrac{\mathrm{m}\,\omega^2}{\mathrm{EI}}\right)^{1/4}$

nato pa še upošteva, da lastna frekvenca ω ni samo ena ter da so rešitve zanje shranjene v spremenljivki z imenom *omege*:

```
In[55]:= B = PowerExpand[B /. ω → omege]
```

$$\text{Out[55]=}\ \left\{\frac{1.8751}{L}, \frac{4.69409}{L}, \frac{7.85476}{L}, \frac{10.9955}{L}, \frac{14.1372}{L}, \frac{17.2788}{L}\right\}$$

Sedaj je mogoče posamezne nihajne oblike tudi izpisati, in npr. prva nihajna oblika je tako:

```
In[56]:= w[x] /. b → B[[1]]
```

$$\text{Out[56]=}\ C[1]\,\mathrm{Cos}\left[\frac{1.8751\,x}{L}\right] - C[1]\,\mathrm{Cosh}\left[\frac{1.8751\,x}{L}\right] - 0.734096\,C[1]\,\mathrm{Sin}\left[\frac{1.8751\,x}{L}\right] + 0.734096\,C[1]\,\mathrm{Sinh}\left[\frac{1.8751\,x}{L}\right]$$

Dobljeno funkcijo je mogoče tudi izrisati, za vrednost dolžine pa je za izris potrebno izbrati neko vrednost, (npr. 1). Ker gre za nihajno obliko, velikost konstante C[1] ni pomembna (vse dokler je od nič različna) in se upošteva npr. kot 1. Prva nihajna oblika je tako:

```
In[57]:= Plot[w[x] /. {C[1] → 1, b → B[[1]]} /. L → 1, {x, 0, 1}];
```




S primerno *Do[]* zanko pa je mogoče celo izpisati (z ukazom *Print[argument]*) in izrisati nekaj prvih lastnih nihajnih oblik (prikazane so zgolj prve tri nihajne oblike, čeprav prikazani niz ukazov dejansko izvede izris vseh prvi šest nihajnih oblik):

In[58]:=
```
Do[Print[w[x] /. b → B[[i]]];
  Plot[w[x] /. {C[1] → 1, b → B[[i]]} /. L → 1, {x, 0, 1}],
  {i, 1, 6}];
```

$$C[1]\,\mathrm{Cos}\left[\frac{1.8751\,x}{L}\right] - C[1]\,\mathrm{Cosh}\left[\frac{1.8751\,x}{L}\right] - 0.734096\,C[1]\,\mathrm{Sin}\left[\frac{1.8751\,x}{L}\right] + 0.734096\,C[1]\,\mathrm{Sinh}\left[\frac{1.8751\,x}{L}\right]$$




$$C[1]\,\mathrm{Cos}\left[\frac{4.69409\,x}{L}\right] - C[1]\,\mathrm{Cosh}\left[\frac{4.69409\,x}{L}\right] - 1.01847\,C[1]\,\mathrm{Sin}\left[\frac{4.69409\,x}{L}\right] + 1.01847\,C[1]\,\mathrm{Sinh}\left[\frac{4.69409\,x}{L}\right]$$



$$C[1]\,\mathrm{Cos}\left[\frac{7.85476\,x}{L}\right] - C[1]\,\mathrm{Cosh}\left[\frac{7.85476\,x}{L}\right] - 0.999224\,C[1]\,\mathrm{Sin}\left[\frac{7.85476\,x}{L}\right] + 0.999224\,C[1]\,\mathrm{Sinh}\left[\frac{7.85476\,x}{L}\right]$$




**NALOGE ZA SAMOSTOJNO DELO**

1. Poišči karakteristično enačbo ter lastne frekvence obojestransko polnovpetega nosilca dolžine L

REŠITVE

$$2\cdot b^2\cdot(-1+\mathrm{Cos}(b\cdot L)\cdot \mathrm{Cosh}(b\cdot L))$$

$$\omega_1=\frac{22.3733}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_2=\frac{61.6728}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_3=\frac{120.903}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_4=\frac{199.859}{L^2}\cdot\sqrt{\frac{EI}{m}}$$

$$\omega_5=\frac{416.991}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_6=\frac{555.165}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_7=\frac{713.079}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_8=\frac{890.732}{L^2}\cdot\sqrt{\frac{EI}{m}}$$

2. Poišči karakteristično enačbo ter lastne frekvence nosilca dolžine L, ki je na levi polno vpet, na desni pa členkasto priključen

REŠITVE

$$2\cdot b^3\cdot(-\mathrm{Cosh}(b\cdot L)\cdot \mathrm{Sin}(b\cdot L)+\mathrm{Cos}(b\cdot L)\cdot \mathrm{Sinh}(b\cdot L))$$

$$\omega_1=\frac{15.4182}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_2=\frac{49.9649}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_3=\frac{104.248}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_4=\frac{178.27}{L^2}\cdot\sqrt{\frac{EI}{m}}$$

$$\omega_5=\frac{272.031}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_6=\frac{385.531}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_7=\frac{518.771}{L^2}\cdot\sqrt{\frac{EI}{m}}\quad \omega_8=\frac{671.75}{L^2}\cdot\sqrt{\frac{EI}{m}}$$